The Stark Effect on Hydrogen for n=3 and Transitions from n=3 to n=2
Joseph Jung and Bobby Barjasteh
Quantum Mechanics Final Project

Constants

```
h = 4.135667662 * 10^-15;
c = 2.99792458 * 10^8;
nair = 1.000276;
qe = -1.602177 * 10^-19;
α = 5.29 * 10^-11;
```

Define the radial functions for Hydrogen (n = 3)

$$R[3, 0, r\_] = \frac{2}{\sqrt{27}} a_0^{-3/2} \left(1 - \frac{2}{3} \frac{r}{a_0} + \frac{2}{27} \left(\frac{r}{a_0}\right)^2\right) \mathrm{Exp}\left[\frac{-r}{3 a_0}\right];$$

$$R[3, 1, r\_] = \frac{8}{27 \sqrt{6}} a_0^{-3/2} \left(1 - \frac{1}{6} \frac{r}{a_0}\right) \left(\frac{r}{a_0}\right) \mathrm{Exp}\left[\frac{-r}{3 a_0}\right];$$

$$R[3, 2, r\_] = \frac{4}{81 \sqrt{30}} a_0^{-3/2} \left(\frac{r}{a_0}\right)^2 \mathrm{Exp}\left[\frac{-r}{3 a_0}\right];$$

Wave functions for Hydrogen

```
ψ[n_, ℓ_, m_, r_, θ_, ϕ_] = R[n, ℓ, r] * SphericalHarmonicY[ℓ, m, θ, ϕ];
```

Perturbation

```
H1[r_, θ_, ϕ_] = e * E0 * r * Cos[θ];
```

Perturbation matrix elements

```
H[n_, l_, m_, n2_, l2_, m2_] := Integrate[Conjugate[ψ[n, l, m, r, θ, ϕ]] * H1[r, θ, ϕ] *
    ψ[n2, l2, m2, r, θ, ϕ] * r^2 * Sin[θ], {r, 0, ∞}, {θ, 0, π}, {ϕ, 0, 2 π}];
```

Define the states (n, *l*, m)

```
(states = {{3, 0, 0}, {3, 1, 0}, {3, 1, 1}, {3, 1, -1},
    {3, 2, 0}, {3, 2, 1}, {3, 2, 2}, {3, 2, -1}, {3, 2, -2}}) // MatrixForm
```

$$\begin{pmatrix} 3 & 0 & 0 \\ 3 & 1 & 0 \\ 3 & 1 & 1 \\ 3 & 1 & -1 \\ 3 & 2 & 0 \\ 3 & 2 & 1 \\ 3 & 2 & 2 \\ 3 & 2 & -1 \\ 3 & 2 & -2 \end{pmatrix}$$

```
psiStates = {ψ300, ψ310, ψ311, ψ3101, ψ320, ψ321, ψ322, ψ3201, ψ3202};
psiStateFuncs[r_, θ_] = {ψ[3, 0, 0, r, θ, ϕ], ψ[3, 1, 0, r, θ, ϕ],
   ψ[3, 1, 1, r, θ, ϕ], ψ[3, 1, -1, r, θ, ϕ], ψ[3, 2, 0, r, θ, ϕ], ψ[3, 2, 1, r, θ, ϕ],
   ψ[3, 2, 2, r, θ, ϕ], ψ[3, 2, -1, r, θ, ϕ], ψ[3, 2, -2, r, θ, ϕ]};
```

Compute the perturbation matrix (for first order corrections to energy)

```
energyMatrix := Table[Table[
    H[states[[i, 1]], states[[i, 2]], states[[i, 3]], states[[j, 1]], states[[j, 2]],
     states[[j, 3]]], {j, 1, Length[states]}], {i, 1, Length[states]}];
```

```
energyMatrixSimplified = Assuming[a0 > 0, Simplify[energyMatrix]];
energyMatrixWithLabels = Prepend[energyMatrixSimplified,
   {"300", "310", "311", "31-1", "320", "321", "322", "32-1", "32-2"}];
energyMatrixWithLabels = MapThread[Prepend, {energyMatrixWithLabels,
    {"", "300", "310", "311", "31-1", "320", "321", "322", "32-1", "32-2"}}];
Grid[energyMatrixWithLabels, Frame → All]
```

	300	310	311	31-1	320	321	322	32-1	32-2
300	0	$-3\sqrt{6}\ e\ E0\ a_0$	0	0	0	0	0	0	0
310	$-3\sqrt{6}\ e\ E0\ a_0$	0	0	0	$-3\sqrt{3}\ e\ E0\ a_0$	0	0	0	0
311	0	0	0	0	0	$-\frac{9}{2}\ e\ E0\ a_0$	0	0	0
31-1	0	0	0	0	0	0	0	$-\frac{9}{2}\ e\ E0\ a_0$	0
320	0	$-3\sqrt{3}\ e\ E0\ a_0$	0	0	0	0	0	0	0
321	0	0	$-\frac{9}{2}\ e\ E0\ a_0$	0	0	0	0	0	0
322	0	0	0	0	0	0	0	0	0
32-1	0	0	0	$-\frac{9}{2}\ e\ E0\ a_0$	0	0	0	0	0
32-2	0	0	0	0	0	0	0	0	0

Determine the eigenvalues and eigenvectors of the perturbation matrix

The eiegenvectors of the perturbation matrix reveal the mixed states which we use as our new bases. Some of the degeneracies are lifted.

```
(system = Transpose[Eigensystem[energyMatrixSimplified]]) // MatrixForm
```

$$\begin{pmatrix}
-9\ e\ E0\ a_0 & \{\sqrt{2}, \sqrt{3}, 0, 0, 1, 0, 0, 0, 0\} \\
9\ e\ E0\ a_0 & \{\sqrt{2}, -\sqrt{3}, 0, 0, 1, 0, 0, 0, 0\} \\
-\frac{9}{2}\ e\ E0\ a_0 & \{0, 0, 0, 1, 0, 0, 0, 1, 0\} \\
-\frac{9}{2}\ e\ E0\ a_0 & \{0, 0, 1, 0, 0, 1, 0, 0, 0\} \\
\frac{9}{2}\ e\ E0\ a_0 & \{0, 0, 0, -1, 0, 0, 0, 1, 0\} \\
\frac{9}{2}\ e\ E0\ a_0 & \{0, 0, -1, 0, 0, 1, 0, 0, 0\} \\
0 & \{0, 0, 0, 0, 0, 0, 0, 0, 1\} \\
0 & \{0, 0, 0, 0, 0, 0, 1, 0, 0\} \\
0 & \{-\frac{1}{\sqrt{2}}, 0, 0, 0, 1, 0, 0, 0, 0\}
\end{pmatrix}$$

The eigenvectors written as linear combinations of ψ states

```
(systemWithStates = Table[{system[[i, 1]], Normalize[system[[i, 2]]].psiStates},
   {i, 1, Length[psiStates]}]) // MatrixForm
```

$$\begin{pmatrix} -9\,\mathrm{e}\,\mathrm{E0}\,a_0 & \frac{\psi_{300}}{\sqrt{3}} + \frac{\psi_{310}}{\sqrt{2}} + \frac{\psi_{320}}{\sqrt{6}} \\ 9\,\mathrm{e}\,\mathrm{E0}\,a_0 & \frac{\psi_{300}}{\sqrt{3}} - \frac{\psi_{310}}{\sqrt{2}} + \frac{\psi_{320}}{\sqrt{6}} \\ -\frac{9}{2}\,\mathrm{e}\,\mathrm{E0}\,a_0 & \frac{\psi_{3101}}{\sqrt{2}} + \frac{\psi_{3201}}{\sqrt{2}} \\ -\frac{9}{2}\,\mathrm{e}\,\mathrm{E0}\,a_0 & \frac{\psi_{311}}{\sqrt{2}} + \frac{\psi_{321}}{\sqrt{2}} \\ \frac{9}{2}\,\mathrm{e}\,\mathrm{E0}\,a_0 & -\frac{\psi_{3101}}{\sqrt{2}} + \frac{\psi_{3201}}{\sqrt{2}} \\ \frac{9}{2}\,\mathrm{e}\,\mathrm{E0}\,a_0 & -\frac{\psi_{311}}{\sqrt{2}} + \frac{\psi_{321}}{\sqrt{2}} \\ 0 & \psi_{3202} \\ 0 & \psi_{322} \\ 0 & -\frac{\psi_{300}}{\sqrt{3}} + \sqrt{\frac{2}{3}}\,\psi_{320} \end{pmatrix}$$

Degeneracies:
3 states with ΔE = 0
2 states with ΔE = 9/2 e E0 a0
2 states with ΔE = -9/2 e E0 a0

```
test[r_, θ_] = (1 / Sqrt[3]) ψ[3, 0, 0, r, θ, ϕ] +
   (1 / Sqrt[2]) ψ[3, 1, 0, r, θ, ϕ] + (1 / Sqrt[6]) ψ[3, 2, 0, r, θ, ϕ];
```

Substituting the actual forms of the wave functions:

```
(systemWithFuncs[r_, θ_] =
   Table[{system[[i, 1]], Normalize[system[[i, 2]]].psiStateFuncs[r, θ]},
    {i, 1, Length[psiStates]}]) // MatrixForm
```

$$\begin{pmatrix}
-9\,e\,E0\,a_0 & \frac{e^{-\frac{r}{3a_0}} r^2 \left(-1+3\,\mathrm{Cos}[\theta]^2\right)}{486\sqrt{\pi}\,a_0^{7/2}} + \frac{2\,e^{-\frac{r}{3a_0}} r\,\mathrm{Cos}[\theta]\left(1-\frac{r}{6a_0}\right)}{27\sqrt{\pi}\,a_0^{5/2}} + \frac{e^{-\frac{r}{3a_0}}\left(1+\frac{2r^2}{27a_0^2}-\frac{2r}{3a_0}\right)}{9\sqrt{\pi}\,a_0^{3/2}} \\
9\,e\,E0\,a_0 & \frac{e^{-\frac{r}{3a_0}} r^2 \left(-1+3\,\mathrm{Cos}[\theta]^2\right)}{486\sqrt{\pi}\,a_0^{7/2}} - \frac{2\,e^{-\frac{r}{3a_0}} r\,\mathrm{Cos}[\theta]\left(1-\frac{r}{6a_0}\right)}{27\sqrt{\pi}\,a_0^{5/2}} + \frac{e^{-\frac{r}{3a_0}}\left(1+\frac{2r^2}{27a_0^2}-\frac{2r}{3a_0}\right)}{9\sqrt{\pi}\,a_0^{3/2}} \\
-\frac{9}{2}\,e\,E0\,a_0 & \frac{e^{-i\phi-\frac{r}{3a_0}} r^2\,\mathrm{Cos}[\theta]\,\mathrm{Sin}[\theta]}{81\sqrt{2}\sqrt{\pi}\,a_0^{7/2}} + \frac{e^{-i\phi-\frac{r}{3a_0}}\sqrt{\frac{2}{\pi}}\,r\,\mathrm{Sin}[\theta]\left(1-\frac{r}{6a_0}\right)}{27\,a_0^{5/2}} \\
-\frac{9}{2}\,e\,E0\,a_0 & -\frac{e^{i\phi-\frac{r}{3a_0}} r^2\,\mathrm{Cos}[\theta]\,\mathrm{Sin}[\theta]}{81\sqrt{2}\sqrt{\pi}\,a_0^{7/2}} - \frac{e^{i\phi-\frac{r}{3a_0}}\sqrt{\frac{2}{\pi}}\,r\,\mathrm{Sin}[\theta]\left(1-\frac{r}{6a_0}\right)}{27\,a_0^{5/2}} \\
\frac{9}{2}\,e\,E0\,a_0 & \frac{e^{-i\phi-\frac{r}{3a_0}} r^2\,\mathrm{Cos}[\theta]\,\mathrm{Sin}[\theta]}{81\sqrt{2}\sqrt{\pi}\,a_0^{7/2}} - \frac{e^{-i\phi-\frac{r}{3a_0}}\sqrt{\frac{2}{\pi}}\,r\,\mathrm{Sin}[\theta]\left(1-\frac{r}{6a_0}\right)}{27\,a_0^{5/2}} \\
\frac{9}{2}\,e\,E0\,a_0 & -\frac{e^{i\phi-\frac{r}{3a_0}} r^2\,\mathrm{Cos}[\theta]\,\mathrm{Sin}[\theta]}{81\sqrt{2}\sqrt{\pi}\,a_0^{7/2}} + \frac{e^{i\phi-\frac{r}{3a_0}}\sqrt{\frac{2}{\pi}}\,r\,\mathrm{Sin}[\theta]\left(1-\frac{r}{6a_0}\right)}{27\,a_0^{5/2}} \\
0 & \frac{e^{-2i\phi-\frac{r}{3a_0}} r^2\,\mathrm{Sin}[\theta]^2}{162\sqrt{\pi}\,a_0^{7/2}} \\
0 & \frac{e^{2i\phi-\frac{r}{3a_0}} r^2\,\mathrm{Sin}[\theta]^2}{162\sqrt{\pi}\,a_0^{7/2}} \\
0 & \frac{e^{-\frac{r}{3a_0}} r^2\left(-1+3\,\mathrm{Cos}[\theta]^2\right)}{243\sqrt{\pi}\,a_0^{7/2}} - \frac{e^{-\frac{r}{3a_0}}\left(1+\frac{2r^2}{27a_0^2}-\frac{2r}{3a_0}\right)}{9\sqrt{\pi}\,a_0^{3/2}}
\end{pmatrix}$$

Expectation value of z

```
(expectedz =
   Table[Assuming[a0 > 0 && ϕ ∈ Reals, Integrate[systemWithFuncs[r, θ][[i, 2]]*
        systemWithFuncs[r, θ][[i, 2]] * r * Cos[θ] * r^2 * Sin[θ], {θ, 0, π}, {r, 0, ∞}] *
      2 * π], {i, 1, Length[psiStates]}]) // MatrixForm
```

$$\begin{pmatrix}
-9\,a_0 \\
9\,a_0 \\
-\frac{9\,a_0}{2} \\
-\frac{9\,a_0}{2} \\
\frac{9\,a_0}{2} \\
\frac{9\,a_0}{2} \\
0 \\
0 \\
0
\end{pmatrix}$$

```
(*g[i_,r_,θ_]:=Simplify[
  systemWithFuncs[r,θ][[i,2]]*systemWithFuncs[r,θ][[i,2]]*r^2*Sin[θ]/.a0→α];
Quiet@Show[ContourPlot[g[2,Sqrt[x^2+y^2],ArcTan[x,y]],
    {x,0,50α},{y,0,50α},Contours→10,AspectRatio→1]]
 Quiet@Show[ContourPlot[g[2,Sqrt[x^2+y^2],ArcTan[x,y]],
    {x,-50α,0},{y,0,50α},Contours→10,AspectRatio→1]]
 Quiet@Show[ContourPlot[g[2,Sqrt[x^2+y^2],ArcTan[x,y]],
    {x,0,50α},{y,-50α,0},Contours→10,AspectRatio→1]]
 Quiet@Show[ContourPlot[g[2,Sqrt[x^2+y^2],ArcTan[x,y]],
    {x,-50α,0},{y,-50α,0},Contours→10,AspectRatio→1]]*)
```

The unique shifts in energy due to the Stark Effect for n = 3

```
(n3StarkShifts = Reverse[DeleteDuplicates[Transpose[systemWithStates][[1]]]]) //
 MatrixForm
```

$$\begin{pmatrix} 0 \\ \frac{9}{2}\, e\, E0\, a_0 \\ -\frac{9}{2}\, e\, E0\, a_0 \\ 9\, e\, E0\, a_0 \\ -9\, e\, E0\, a_0 \end{pmatrix}$$

The normalized eigenvectors resulting from the Stark Effect for n = 2 as done in class

```
(n2system = {{-3 * e * E0 * a0, (1 / Sqrt[2]) (ψ200 + ψ210)},
    {3 * e * E0 * a0, (1 / Sqrt[2]) (-ψ200 + ψ210)}, {0, ψ211}, {0, ψ2101}}) // MatrixForm
```

$$\begin{pmatrix} -3\, e\, E0\, a_0 & \frac{\psi_{200}+\psi_{210}}{\sqrt{2}} \\ 3\, e\, E0\, a_0 & \frac{-\psi_{200}+\psi_{210}}{\sqrt{2}} \\ 0 & \psi_{211} \\ 0 & \psi_{2101} \end{pmatrix}$$

The unique shifts in energy due to the Stark Effect for n = 2

```
(n2StarkShifts = Reverse[DeleteDuplicates[Transpose[n2system][[1]]]]) // MatrixForm
```

$$\begin{pmatrix} 0 \\ 3\, e\, E0\, a_0 \\ -3\, e\, E0\, a_0 \end{pmatrix}$$

Now, we compute the differences in energy

Define the external field strength (in units of eV / C * m)

```
Eext = 1 000 000 / qe;
varset = {e → qe, a0 → α, E0 → Eext, n → nair};
```

First, the energy difference due to the unperturbed Bohr energy will be constant for each of the transitions

```
EBohr[n_] = -13.598 / n^2;
```

```
ΔEBohr = EBohr[3] - EBohr[2]
```

```
1.88861
```

Then, the energy of the emitted photon will also have an additional amount corresponding to the differences in the shifts due to the Stark effect of the initial and final states. We label the states from 1-5 for n = 3 (there are 5 unique eigenvalues. The degeneracies do not change the energy shifts) and 1-3 for n = 2 (there are 3 unique eigenvalues).

```
ΔEStark[m3_, m2_] := n3StarkShifts[[m3]] - n2StarkShifts[[m2]];
```

```
ΔETotal[m3_, m2_] := ΔEBohr + ΔEStark[m3, m2];
```

The frequency of the emitted photon is given by:

```
νPhoton[En_] = En / h;
```

Traveling through a medium with a given index of refraction, the wavelength is given by:

```
λPhoton[En_, n_] = c / (νPhoton[En] * n) * 10^9;
```

Table of energy differences

```
transition[m3_] :=
  Table[{m3, m2, ΔEStark[m3, m2], ΔETotal[m3, m2], νPhoton[ΔETotal[m3, m2]],
     λPhoton[ΔETotal[m3, m2], n]}, {m2, 1, Length[n2StarkShifts]}] /. varset;
transitions = {{"k(3)", "k(2)", "ΔE Stark", "ΔE Total", "ν (Hz)", "λ (nm)"}};
For[m3 = 1, m3 ≤ Length[n3StarkShifts],
  m3++, transitions = Join[transitions, transition[m3]]];
Grid[transitions, Frame → All]
```

k(3)	k(2)	ΔE Stark	ΔE Total	ν (Hz)	λ (nm)
1	1	0	1.88861	4.56664×10^{14}	656.302
1	2	-0.0001587	1.88845	4.56626×10^{14}	656.357
1	3	0.0001587	1.88877	4.56703×10^{14}	656.247
2	1	0.00023805	1.88885	4.56722×10^{14}	656.22
2	2	0.00007935	1.88869	4.56683×10^{14}	656.275
2	3	0.00039675	1.88901	4.5676×10^{14}	656.164
3	1	-0.00023805	1.88837	4.56607×10^{14}	656.385
3	2	-0.00039675	1.88821	4.56568×10^{14}	656.44
3	3	-0.00007935	1.88853	4.56645×10^{14}	656.33
4	1	0.0004761	1.88909	4.56779×10^{14}	656.137
4	2	0.0003174	1.88893	4.56741×10^{14}	656.192
4	3	0.0006348	1.88925	4.56818×10^{14}	656.082
5	1	-0.0004761	1.88814	4.56549×10^{14}	656.468
5	2	-0.0006348	1.88798	4.56511×10^{14}	656.523
5	3	-0.0003174	1.88829	4.56587×10^{14}	656.413

Resulting wavelengths (sorted from smallest to largest)

```
(wavelengths = Sort[Transpose[transitions][[-1, 2 ;;]]]) // MatrixForm
```

$$\begin{pmatrix} 656.082 \\ 656.137 \\ 656.164 \\ 656.192 \\ 656.22 \\ 656.247 \\ 656.275 \\ 656.302 \\ 656.33 \\ 656.357 \\ 656.385 \\ 656.413 \\ 656.44 \\ 656.468 \\ 656.523 \end{pmatrix}$$

If the Stark effect and other effects are ignored, we would expect a wavelength (nm) of:

```
c * h / ((EBohr[3] - EBohr[2]) * nair) * 10^9
```

656.302